Does $\sum_{k=1}^{\infty} \frac{1}{8^k+2^k+1}$ have a closed form? Is there a closed form for this infinite summation?
$$\sum_{k=1}^{\infty} \frac{1}{8^k+2^k+1}$$
 A: Let $x=2^k$. Then let $\alpha_1$,$\alpha_2$ and $\alpha_3$ be roots of $x^3 + x+1$ and let 
$$
   \frac{1}{x^3+x+1} = \sum_{m=1}^3 \frac{\omega_m}{x+\alpha_m} 
$$
be the partial fraction decomposition. Then
$$
  \sum_{k=1}^\infty \frac{1}{8^k+2^k+1} = \sum_{m=1}^3 \omega_m \sum_{k=1}^\infty \frac{1}{2^k+\alpha_m}  = -1 + \sum_{m=1}^3 \frac{\omega_m}{\alpha_m \log(2)} \psi_q \left(-\log_2\left(-\frac{\alpha_m}{2}\right); \frac{1}{2}\right)
$$
where $\psi_q$ denotes q-digamma function.

Here is relevant Mathematica computation:

In[270]:= 
pfd = Root[-1 - 3 #1 + 31 #1^3 &, 1]/(x + Root[-1 + #1 + #1^3 &, 1]) +
    Root[-1 - 3 #1 + 31 #1^3 &, 2]/(x + Root[-1 + #1 + #1^3 &, 2]) + 
   Root[-1 - 3 #1 + 31 #1^3 &, 3]/(x + Root[-1 + #1 + #1^3 &, 3]);

In[271]:= pfd == 1/(x^3 + x + 1) // FullSimplify

Out[271]= True

In[274]:= Sum[1/(2^k + r), {k, 1, \[Infinity]}]

Out[274]= (-Log[2] + 
 QPolyGamma[0, -(Log[-(r/2)]/Log[2]), 1/2])/(r Log[2])

In[279]:= 
res = (pfd /. (x + r_)^(-1) :> 
      QPolyGamma[0, -(Log[-(r/2)]/Log[2]), 1/2]/(r Log[2])) - 1;

In[280]:= N[res, 20]

Out[280]= 0.107599207416490556744 + 0.*10^-22 I

In[281]:= N[Sum[1/(8^k + 2^k + 1), {k, 1, 50}], 20]

Out[281]= 0.10759920741649055674

